Linear elasticity

Example from Landau-Lifschitz, Theory of elasticity, volume 7, Pergaom Press, 1970.

Let us introduce the following Ore algebra A of partial differential operators

```
A = OreAlgebra[Der[x], Der[y], Der[z]]
```

$\mathbb{K}(x, y, z)[D_x; 1, D_x][D_y; 1, D_y][D_z; 1, D_z]$

and the matrix R of partial differential operators, which defines the equilibrium of the stress tensor (forces are absent)

```
R = {{Der[x], 0, 0, 0, Der[z], Der[y]},
   {0, Der[y], 0, Der[z], 0, Der[x]},
   {0, 0, Der[z], Der[y], Der[x], 0}};
MatrixForm[R = ToOrePolynomial[R, A]]
```

$$\begin{pmatrix} D_x & 0 & 0 & 0 & D_z & D_y \\ 0 & D_y & 0 & D_z & 0 & D_x \\ 0 & 0 & D_z & D_y & D_x & 0 \end{pmatrix}$$

and the left A-module, finitely presented by R, defined by M = $A^{1\times 6} / A^{1\times 3} R$.

Let us write the equations of the system explicitly:

```
u = {x, y, z};
EMF = {σ1 @@ u, σ2 @@ u, σ3 @@ u, τ1 @@ u, τ2 @@ u, τ3 @@ u}
```

$\{\sigma_1[x, y, z], \sigma_2[x, y, z], \sigma_3[x, y, z], \tau_1[x, y, z], \tau_2[x, y, z], \tau_3[x, y, z]\}$

```
sys = Thread[ApplyMatrix[R, EMF] == 0]
```

$$\{\tau_2{}^{(0,0,1)}[x, y, z] + \tau_3{}^{(0,1,0)}[x, y, z] + \sigma_1{}^{(1,0,0)}[x, y, z] == 0,$$
$$\tau_1{}^{(0,0,1)}[x, y, z] + \sigma_2{}^{(0,1,0)}[x, y, z] + \tau_3{}^{(1,0,0)}[x, y, z] == 0,$$
$$\sigma_3{}^{(0,0,1)}[x, y, z] + \tau_1{}^{(0,1,0)}[x, y, z] + \tau_2{}^{(1,0,0)}[x, y, z] == 0\}$$

Let us compute the adjoint of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_x & 0 & 0 \\ 0 & -D_y & 0 \\ 0 & 0 & -D_z \\ 0 & -D_z & -D_y \\ -D_z & 0 & -D_x \\ -D_y & -D_x & 0 \end{pmatrix}$$

Let us check whether or not M is torsion-free:

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & D_z & D_y & D_x & 0 \\ 0 & -D_y & 0 & -D_z & 0 & -D_x \\ -D_x & 0 & 0 & 0 & -D_z & -D_y \end{pmatrix}, \right.$$

$$\left. \begin{pmatrix} 0 & 0 & 0 & D_z^2 & -2\, D_y\, D_z & D_y^2 \\ 0 & D_z^2 & 2\, D_x\, D_z & 0 & 0 & D_x^2 \\ 2\, D_x\, D_y & D_y^2 & 0 & D_x^2 & 0 & 0 \\ -D_x\, D_z & -D_y\, D_z & -D_x\, D_y & 0 & -D_x^2 & 0 \\ -D_y\, D_z & 0 & D_y^2 & -D_x\, D_z & D_x\, D_y & 0 \\ D_z^2 & 0 & -D_y\, D_z & 0 & D_x\, D_z & -D_x\, D_y \end{pmatrix} \right\}$$

Since the first matrix is identity, M is torsion-free and thus the system is parametrized by Q, i.e:

```
Thread[EMF == ApplyMatrix[Q, {θ1 @@ u, θ2 @@ u, θ3 @@ u, ψ1 @@ u, ψ2 @@ u, ψ3 @@ u}]]
```

$$\{\sigma_1[x, y, z] == \psi_1^{(0,0,2)}[x, y, z] - 2\, \psi_2^{(0,1,1)}[x, y, z] + \psi_3^{(0,2,0)}[x, y, z],$$
$$\sigma_2[x, y, z] == \theta_2^{(0,0,2)}[x, y, z] + 2\, \theta_3^{(1,0,1)}[x, y, z] + \psi_3^{(2,0,0)}[x, y, z],$$
$$\sigma_3[x, y, z] == \theta_2^{(0,2,0)}[x, y, z] + 2\, \theta_1^{(1,1,0)}[x, y, z] + \psi_1^{(2,0,0)}[x, y, z],$$
$$\tau_1[x, y, z] == -\theta_2^{(0,1,1)}[x, y, z] - \theta_1^{(1,0,1)}[x, y, z] -$$
$$\theta_3^{(1,1,0)}[x, y, z] - \psi_2^{(2,0,0)}[x, y, z],\ \tau_2[x, y, z] ==$$
$$-\theta_1^{(0,1,1)}[x, y, z] + \theta_3^{(0,2,0)}[x, y, z] - \psi_1^{(1,0,1)}[x, y, z] + \psi_2^{(1,1,0)}[x, y, z],$$
$$\tau_3[x, y, z] == \theta_1^{(0,0,2)}[x, y, z] - \theta_3^{(0,1,1)}[x, y, z] + \psi_2^{(1,0,1)}[x, y, z] - \psi_3^{(1,1,0)}[x, y, z]\}$$

We note the above parametrization involves six arbitrary functions θ_1, θ_2, θ_3, ψ_1, ψ_2, ψ_3. Let us now compute the rank of the system

```
OreRank[R, A]
```

3

From that we know that there exist parametrizations involving only three arbitrary functions. Let us compute two of them leading to standard parametrizations:

```
MatrixForm[Q1 = #[[{1, 3, 5}]] & /@ Q]
```

$$\begin{pmatrix} 0 & 0 & -2\, D_y\, D_z \\ 0 & 2\, D_x\, D_z & 0 \\ 2\, D_x\, D_y & 0 & 0 \\ -D_x\, D_z & -D_x\, D_y & -D_x^2 \\ -D_y\, D_z & D_y^2 & D_x\, D_y \\ D_z^2 & -D_y\, D_z & D_x\, D_z \end{pmatrix}$$

```
ColumnForm@ApplyMatrix[Q1, {ψ3[x, y, z] / 2, ψ2[x, y, z] / 2, -ψ1[x, y, z] / 2}]
```

$$\psi_1^{(0,1,1)}[x, y, z]$$
$$\psi_2^{(1,0,1)}[x, y, z]$$
$$\psi_3^{(1,1,0)}[x, y, z]$$
$$\frac{1}{2}\left(-\psi_3^{(1,0,1)}[x, y, z] - \psi_2^{(1,1,0)}[x, y, z] + \psi_1^{(2,0,0)}[x, y, z]\right)$$
$$\frac{1}{2}\left(-\psi_3^{(0,1,1)}[x, y, z] + \psi_2^{(0,2,0)}[x, y, z] - \psi_1^{(1,1,0)}[x, y, z]\right)$$
$$\frac{1}{2}\left(\psi_3^{(0,0,2)}[x, y, z] - \psi_2^{(0,1,1)}[x, y, z] - \psi_1^{(1,0,1)}[x, y, z]\right)$$

This is the Morera parametrization.

```
MatrixForm[Q2 = #[[{2, 4, 6}]] & /@ Q]
```

$$\begin{pmatrix} 0 & D_z^2 & D_y^2 \\ D_z^2 & 0 & D_x^2 \\ D_y^2 & D_x^2 & 0 \\ -D_y D_z & 0 & 0 \\ 0 & -D_x D_z & 0 \\ 0 & 0 & -D_x D_y \end{pmatrix}$$

```
ColumnForm@ApplyMatrix[Q2, {θ1[x, y, z], θ2[x, y, z], θ3[x, y, z]}]
```

$\theta_2{}^{(0,0,2)}[x, y, z] + \theta_3{}^{(0,2,0)}[x, y, z]$

$\theta_1{}^{(0,0,2)}[x, y, z] + \theta_3{}^{(2,0,0)}[x, y, z]$

$\theta_1{}^{(0,2,0)}[x, y, z] + \theta_2{}^{(2,0,0)}[x, y, z]$

$-\theta_1{}^{(0,1,1)}[x, y, z]$

$-\theta_2{}^{(1,0,1)}[x, y, z]$

$-\theta_3{}^{(1,1,0)}[x, y, z]$

This is the Maxwell parametrization.